Supplement of Magn. Reson., 2, 511–522, 2021
https://doi.org/10.5194/mr-2-511-2021-supplement

Open Access
MAGNETIC
RESONANCE

Supplement of

Exclusively heteronuclear NMR experiments for the investigation of intrinsically disordered proteins: focusing on proline residues

Isabella C. Felli et al.

Correspondence to: Isabella C. Felli (felli@cerm.unifi.it) and Roberta Pierattelli (roberta.pierattelli@unifi.it)

Supplementary information

The pulse sequences used to acquire the 3D experiments to focus on proline residues are reported hereafter:

Pulse sequence 1 – **3D (H)CBCACON_pro**

Pulse sequence 2 – **3D (H)CCCON_pro**

Pulse sequence 3 – **3D (H)CBCANCO_pro**

Pulse sequence 4 – **3D (H)CACOCON_pro**

Pulse sequence 1 – **3D (H)CBCACON_pro**

```
;c_hcbcacon_ia3d_pro
;avance-version (20/07/28)
;(H)CbCaCON
;3D sequence with
;   13C detected correlation for triple resonance using
;      multiple inept transfer steps
;
;      F2(Ha/b) -> F1(Ca/b,t1) -> F1(Ca) -> F1(C=O)
;               -> F3(N,t2) -> F1(C=O,t3)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;phase sensitive (t2)
;using constant time in t1
;using IPAP scheme for virtual decoupling
;using selective N pulse for Pro
;(use parameterset )
;
;M.G. Murrali, A. Piai, W. Bermel, I.C. Felli & R. Pierattelli,
;   ChemBioChem 19, 1625-1629 (2018)
;W. Bermel, I. Bertini, V. Csizmok, I. C. Felli, R. Pierattelli &
;   P. Tompa, J. Magn. Reson. 198, 275-281 (2009)
;(W. Bermel, I. Bertini, I.C. Felli, R. Kuemmerle
;   & R. Pierattelli, J. Magn. Reson. 178, 56-64 (2006) )
;(W. Bermel, I. Bertini,L. Duma, I.C. Felli, L. Emsley,  R. Pierattelli,
;   P.R. Vasos, Angew. Chem. Int. Ed. 44, 3089-3092 (2005) )
;(L. Duma, S. Hediger, A. Lesage & L. Emsley,
;   J. Magn. Reson. 164, 187-195 (2003) )
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple_c>


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>


"p4=p3*2"
```

```
"p22=p21*2"
"d11=30m"
"d12=20u"

"d3=1.1m"
"d4=1.8m"
"d22=4.5m"
"d23=16.6m"
"d25=4.0m"


"d0=3u"
"d10=3u"
"d20=d25-p12-4u"

"in0=inf1/2"
"in10=inf2/2"

"in20=in0"

"td1=tdmax(td1,d20*2,in20)"


"DELTA=d10*2+larger(p4,p8)"
"DELTA1=d25-d0-larger(p12,p22)-d3-p4"
"DELTA2=d23-d22-p12"
"DELTA3=d23-p12-larger(p12,p59)/2-4u"
"DELTA4=d23/2-p12/2"
"DELTA5=d25-p12-4u"
"DELTA6=d23-d22-p12-larger(p12,p59)/2"


"l0=1"


"spoff13=bf1*((cnst22/2-cnst21/2)/1000000)"
"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst21-cnst23)/1000000)"
"spoff27=bf1*((cnst22-cnst21)/1000000)"

"o1_F1=bf1*cnst23/1000000"

aqseq 321


1 ze
```

```
  d11 pl12:f2 pl16:f3
2 d11 do:f2 do:f3
3 d1 fq=cnst23(bf ppm):f1
  d12 pl2:f2 pl3:f3
  50u UNBLKGRAD

  (p3 ph1):f2
  d4
  (center (p12:sp24 ph1) (p4 ph1):f2 )
  d4
  (p3 ph2):f2

  p16:gp1
  d16

  (p11:sp23 ph3)
  d0
  (center (p12:sp26 ph6) (p22 ph1):f3 )
  d3
  (p4 ph6):f2
  DELTA1
  (p12:sp24 ph1)
  d20
  (p12:sp26 ph1)
  4u
  (p11:sp25 ph1)

  4u
  (p12:sp26 ph1)
  DELTA5
  (p12:sp24 ph1)
  4u
  (p12:sp26 ph1)
  DELTA5
  (p11:sp23 ph1)

  p16:gp2
  d16 fq=cnst21(bf ppm):f1

  (p11:sp23 ph5)
  d22
  (p12:sp27 ph1)
  DELTA6
  (center (p12:sp24 ph1) (p59:sp46 ph1):f3 )
  DELTA3
  (p12:sp27 ph1)
  4u
  (p11:sp25 ph1)
```

```
151
152     p16:gp3
153     d16 pl3:f3
154
155     (p21 ph4):f3
156     d10
157     (center (p8:sp13 ph6) (p4 ph6):f2 )
158     d10
159     (p22 ph1):f3
160     DELTA
161     (p21 ph1):f3
162
163     p16:gp4
164     d16 pl12:f2
165
166     if "l0 %2 == 1"
167       {
168       (p11:sp23 ph1)
169       DELTA4
170       (p12:sp27 ph1)
171       DELTA4
172       (center (p12:sp24 ph1) (p22 ph1):f3 )
173       DELTA4
174       (p12:sp27 ph1)
175       DELTA4 pl16:f3
176       }
177     else
178       {
179       (p11:sp23 ph7)
180       d22
181       (p12:sp27 ph1)
182       DELTA2
183       (center (p12:sp24 ph1) (p22 ph1):f3 )
184       DELTA4
185       DELTA4 pl16:f3
186       (p12:sp27 ph1)
187       }
188
189     4u BLKGRAD
190     go=2 ph31 cpd2:f2 cpd3:f3
191     d11 do:f2 do:f3 mc #0 to 2
192
193   #   ifdef LABEL_F1
194       F1I(iu0, 2)
195   #   else
196       F2I(iu0, 2)
197   #   endif /*LABEL_F1*/
198
```

```
199         F1PH(calph(ph3, +90), caldel(d0, +in0) & caldel(d20, -in20))
200         F2PH(calph(ph4, +90), caldel(d10, +in10))
201     exit
202
203
204     ph1=0
205     ph2=1
206     ph3=0 2
207     ph4=0 0 2 2
208     ph5=0 0 0 0 2 2 2 2
209     ph6=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
210     ph7=3
211     ph31=0 2 2 0 2 0 0 2
212
213
214     ;pl1 : f1 channel - power level for pulse (default)
215     ;pl2 : f2 channel - power level for pulse (default)
216     ;pl3 : f3 channel - power level for pulse (default)
217     ;pl12: f2 channel - power level for CPD/BB decoupling
218     ;pl16: f3 channel - power level for CPD/BB decoupling
219     ;sp13: f1 channel - shaped pulse 180 degree  (adiabatic)
220     ;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
221     ;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
222     ;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
223     ;               for time reversed pulse
224     ;sp26: f1 channel - shaped pulse 180 degree  (C=O off resonance)
225     ;sp27: f1 channel - shaped pulse 180 degree  (Ca off resonance)
226     ;sp46: f3 channel - shaped pulse 180 degree  (N, selective for Pro)
227     ;p3 : f2 channel -  90 degree high power pulse
228     ;p4 : f2 channel - 180 degree high power pulse
229     ;p8 : f1 channel - 180 degree shaped pulse for inversion (adiabatic)
230     ;p11: f1 channel -  90 degree shaped pulse
231     ;p12: f1 channel - 180 degree shaped pulse
232     ;p16: homospoil/gradient pulse                   [1 msec]
233     ;p21: f3 channel -  90 degree high power pulse
234     ;p22: f3 channel - 180 degree high power pulse
235     ;p59: f3 channel - 180 degree shaped pulse (N, selective for Pro)
236     ;d0 : incremented delay (F1 in 3D)               [3 usec]
237     ;d1 : relaxation delay; 1-5 * T1
238     ;d3 : 1/(6J(HCa))                          [1.1 msec]
239     ;d4 : 1/(4J(HCa))                          [1.8 msec]
240     ;d10: incremented delay (F2 in 3D)              [3 usec]
241     ;d11: delay for disk I/O                      [30 msec]
242     ;d12: delay for power switching                 [20 usec]
243     ;d16: delay for homospoil/gradient recovery
244     ;d20: decremented delay (F1 in 3D) = d25+d0-p12-4u
245     ;d22: 1/(4J(COCa))                          [4.5 msec]
246     ;d23: 1/(4J(NCO))                           [16.6 msec]
```

```
;d25: 1/(8J'(CaCb))                    [4.0 msec]
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;cnst23: Caliphatic chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)
;l0: flag to switch between inphase and antiphase
;in0: 1/(2 * SW(Ca/b)) =  DW(Ca/b)
;nd0: 2
;in10: 1/(2 * SW(N)) =  DW(N)
;nd10: 2
;in20: = in0
;ns: 16 * n
;ds: >= 32
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or TPPI) in F1   td1 max = n * d20 / in20 (n = 4 or = 2 with LABEL_F2)
;FnMODE: States-TPPI (or TPPI) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence


;for z-only gradients:
;gpz1: 50%
;gpz2: 30%
;gpz3: 19%
;gpz4: 13%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100


                              ;preprocessor-flags-start
;LABEL_F2: to do ipap in F2 start experiment with
;          option -DLABEL_F2 (eda: ZGOPTNS)
                              ;preprocessor-flags-end


;use AU-program splitcomb [ipap 2] (F1 or F2 with LABEL_F2) to process data



;$Id: $

```

Pulse sequence 2 – **3D (H)CCCON_pro**

```
;c_hcccon_ia3d_pro
;avance-version (20/07/28)
;(H)CC(CO)N
;3D sequence with
;   13C detected correlation for triple resonance using
;      multiple inept transfer steps and
;      C-C FLOPSY16 spinlock
;
;      F1(Cali,t1) -> F1(Ca) -> F1(C=O)
;                          -> F3(N,t2) -> F1(C=O,t3)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;phase sensitive (t2)
;using IPAP scheme for virtual decoupling
;using selective N pulse for Pro
;(use parameterset )
;
;M.G. Murrali, A. Piai, W. Bermel, I.C. Felli & R. Pierattelli,
;   ChemBioChem 19, 1625-1629 (2018)
;W. Bermel, I. Bertini, I.C. Felli, R. Kuemmerle
;   & R. Pierattelli, J. Magn. Reson. 178, 56-64 (2006)
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple_c>


#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p4=p3*2"
"p22=p21*2"
"d11=30m"
"d12=20u"

"d3=0.95m"
```

```
"d4=1.8m"
"d22=4.5m"
"d23=16.6m"


"d0=3u"
"d10=3u"

"in0=inf1/2"
"in10=inf2/2"


"DELTA=d10*2+p8"
"DELTA1=d3/2"
"DELTA2=d3/2+p22+p4+d0*2"
"DELTA3=d23-d22-p12"
"DELTA4=d23-p12-larger(p12,p59)/2-4u"
"DELTA5=d23/2-p12/2"
"DELTA6=d23-d22-p12-larger(p12,p59)/2"


"FACTOR2=(d9/(p6*188.448))"
"l1=FACTOR2"


"l0=1"


"spoff13=bf1*((cnst22/2-cnst21/2)/1000000)"
"spoff22=bf1*((cnst21-cnst22)/1000000)"
"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst21-cnst23)/1000000)"
"spoff27=bf1*((cnst22-cnst21)/1000000)"
"spoff28=0"


"o1_F1=bf1*cnst23/1000000"


aqseq 321


1 ze
  d11 pl12:f2 pl16:f3
2 d11 do:f2 do:f3
3 d1 fq=cnst23(bf ppm):f1
```

```
389     d12 pl2:f2 pl3:f3
390     50u UNBLKGRAD
391
392     (p3 ph1):f2
393     d4
394     (center (p12:sp24 ph1) (p4 ph1):f2 )
395     d4
396     (p3 ph8):f2
397
398     (p11:sp23 ph3)
399     d0
400     (p22 ph1):f3
401     DELTA1
402     (p12:sp26 ph1)
403     DELTA1
404     (p4 ph1):f2
405     d0
406     (p12:sp24 ph1)
407     DELTA2
408     (p12:sp26 ph1)
409     DELTA1
410     (p11:sp25 ph1)
411
412     p19:gp1
413     d16 pl10:f1
414                                                     ;begin FLOPSY16
415   4 p6*0.511 ph11
416     p6*1.067 ph12
417     p6*1.822 ph13
418     p6*1.767 ph14
419     p6*1.444 ph15
420     p6*1.767 ph14
421     p6*1.822 ph13
422     p6*1.067 ph12
423     p6*0.511 ph11
424
425     p6*0.511 ph21
426     p6*1.067 ph22
427     p6*1.822 ph23
428     p6*1.767 ph24
429     p6*1.444 ph25
430     p6*1.767 ph24
431     p6*1.822 ph23
432     p6*1.067 ph22
433     p6*0.511 ph21
434
435     p6*0.511 ph21
436     p6*1.067 ph22
```

```
437     p6*1.822 ph23
438     p6*1.767 ph24
439     p6*1.444 ph25
440     p6*1.767 ph24
441     p6*1.822 ph23
442     p6*1.067 ph22
443     p6*0.511 ph21
444
445     p6*0.511 ph11
446     p6*1.067 ph12
447     p6*1.822 ph13
448     p6*1.767 ph14
449     p6*1.444 ph15
450     p6*1.767 ph14
451     p6*1.822 ph13
452     p6*1.067 ph12
453     p6*0.511 ph11
454
455     p6*0.511 ph11
456     p6*1.067 ph12
457     p6*1.822 ph13
458     p6*1.767 ph14
459     p6*1.444 ph15
460     p6*1.767 ph14
461     p6*1.822 ph13
462     p6*1.067 ph12
463     p6*0.511 ph11
464
465     p6*0.511 ph11
466     p6*1.067 ph12
467     p6*1.822 ph13
468     p6*1.767 ph14
469     p6*1.444 ph15
470     p6*1.767 ph14
471     p6*1.822 ph13
472     p6*1.067 ph12
473     p6*0.511 ph11
474
475     p6*0.511 ph21
476     p6*1.067 ph22
477     p6*1.822 ph23
478     p6*1.767 ph24
479     p6*1.444 ph25
480     p6*1.767 ph24
481     p6*1.822 ph23
482     p6*1.067 ph22
483     p6*0.511 ph21
484
```

```
485     p6*0.511 ph21
486     p6*1.067 ph22
487     p6*1.822 ph23
488     p6*1.767 ph24
489     p6*1.444 ph25
490     p6*1.767 ph24
491     p6*1.822 ph23
492     p6*1.067 ph22
493     p6*0.511 ph21
494
495     p6*0.511 ph21
496     p6*1.067 ph22
497     p6*1.822 ph23
498     p6*1.767 ph24
499     p6*1.444 ph25
500     p6*1.767 ph24
501     p6*1.822 ph23
502     p6*1.067 ph22
503     p6*0.511 ph21
504
505     p6*0.511 ph11
506     p6*1.067 ph12
507     p6*1.822 ph13
508     p6*1.767 ph14
509     p6*1.444 ph15
510     p6*1.767 ph14
511     p6*1.822 ph13
512     p6*1.067 ph12
513     p6*0.511 ph11
514
515     p6*0.511 ph11
516     p6*1.067 ph12
517     p6*1.822 ph13
518     p6*1.767 ph14
519     p6*1.444 ph15
520     p6*1.767 ph14
521     p6*1.822 ph13
522     p6*1.067 ph12
523     p6*0.511 ph11
524
525     p6*0.511 ph21
526     p6*1.067 ph22
527     p6*1.822 ph23
528     p6*1.767 ph24
529     p6*1.444 ph25
530     p6*1.767 ph24
531     p6*1.822 ph23
532     p6*1.067 ph22
```

```
533     p6*0.511 ph21
534
535     p6*0.511 ph21
536     p6*1.067 ph22
537     p6*1.822 ph23
538     p6*1.767 ph24
539     p6*1.444 ph25
540     p6*1.767 ph24
541     p6*1.822 ph23
542     p6*1.067 ph22
543     p6*0.511 ph21
544
545     p6*0.511 ph21
546     p6*1.067 ph22
547     p6*1.822 ph23
548     p6*1.767 ph24
549     p6*1.444 ph25
550     p6*1.767 ph24
551     p6*1.822 ph23
552     p6*1.067 ph22
553     p6*0.511 ph21
554
555     p6*0.511 ph11
556     p6*1.067 ph12
557     p6*1.822 ph13
558     p6*1.767 ph14
559     p6*1.444 ph15
560     p6*1.767 ph14
561     p6*1.822 ph13
562     p6*1.067 ph12
563     p6*0.511 ph11
564
565     p6*0.511 ph11
566     p6*1.067 ph12
567     p6*1.822 ph13
568     p6*1.767 ph14
569     p6*1.444 ph15
570     p6*1.767 ph14
571     p6*1.822 ph13
572     p6*1.067 ph12
573     p6*0.511 ph11
574     lo to 4 times l1
575                                                     ;end FLOPSY16
576
577     4u
578     p19:gp2
579     d16 fq=cnst22(bf ppm):f1
580     20u pl12:f2
```

```
20u cpd2:f2

(p11:sp23 ph1)
d22
(p12:sp22 ph1)
4u
(p25:sp28 ph1)
d22
(p12:sp22 ph1)
4u
(p11:sp25 ph2)

4u do:f2
p16:gp3
d16 fq=cnst21(bf ppm):f1
20u cpd2:f2

(p11:sp23 ph1)
d22
(p12:sp27 ph1)
DELTA6
(center (p12:sp24 ph1) (p59:sp46 ph1):f3 )
DELTA4
(p12:sp27 ph1)
4u
(p11:sp25 ph1)

4u do:f2
p16:gp4
d16 pl3:f3
20u cpd2:f2

(p21 ph4):f3
d10
(p8:sp13 ph6)
d10
(p22 ph1):f3
DELTA
(p21 ph5):f3

if "l0 %2 == 1"
  {
  (p11:sp23 ph1)
  DELTA5
  (p12:sp27 ph1)
  DELTA5
  (center (p12:sp24 ph1) (p22 ph1):f3 )
  DELTA5
```

```
629         (p12:sp27 ph1)
630         DELTA5 pl16:f3
631         }
632       else
633         {
634         (p11:sp23 ph7)
635         d22
636         (p12:sp27 ph1)
637         DELTA3
638         (center (p12:sp24 ph1) (p22 ph1):f3 )
639         DELTA5
640         DELTA5 pl16:f3
641         (p12:sp27 ph1)
642         }
643
644       4u BLKGRAD
645       go=2 ph31 cpd3:f3
646       d11 do:f2 do:f3 mc #0 to 2
647
648     #   ifdef LABEL_F1
649         F1I(iu0, 2)
650     #   else
651         F2I(iu0, 2)
652     #   endif /*LABEL_F1*/
653
654         F1PH(calph(ph3, +90), caldel(d0, +in0))
655         F2PH(calph(ph4, +90), caldel(d10, +in10))
656     exit
657
658
659     ph1=0
660     ph2=1
661     ph3=1 3
662     ph4=0 0 0 0 2 2 2 2
663     ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
664     ph6=0
665     ph7=3
666     ph8=1 1 3 3
667
668     ph11=(720) 0
669     ph12=(720) 90
670     ph13=(720) 135
671     ph14=(720) 630
672     ph15=(720) 45
673
674     ph21=(720) 360
675     ph22=(720) 450
676     ph23=(720) 495
```

```
677     ph24=(720) 270
678     ph25=(720) 405
679
680     ph31=0 2 2 0 2 0 0 2 2 0 0 2 0 2 2 0
681
682
683     ;pl1 : f1 channel - power level for pulse (default)
684     ;pl2 : f2 channel - power level for pulse (default)
685     ;pl3 : f3 channel - power level for pulse (default)
686     ;pl10: f1 channel - power level for TOCSY-spinlock
687     ;pl12: f2 channel - power level for CPD/BB decoupling
688     ;pl16: f3 channel - power level for CPD/BB decoupling
689     ;sp13: f1 channel - shaped pulse 180 degree  (adiabatic)
690     ;sp22: f1 channel - shaped pulse 180 degree  (C=O off resonance)
691     ;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
692     ;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
693     ;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
694     ;               for time reversed pulse
695     ;sp26: f1 channel - shaped pulse 180 degree  (C=O off resonance)
696     ;sp27: f1 channel - shaped pulse 180 degree  (Ca off resonance)
697     ;sp28: f1 channel - shaped pulse 180 degree  (Ca on resonance)
698     ;               sp28 requires higher selectivity than sp24
699     ;sp46: f3 channel - shaped pulse 180 degree  (N, selective for Pro)
700     ;p3 : f2 channel -  90 degree high power pulse
701     ;p4 : f2 channel - 180 degree high power pulse
702     ;p6 : f1 channel -  90 degree low power pulse
703     ;p8 : f1 channel - 180 degree shaped pulse for inversion (adiabatic)
704     ;p11: f1 channel -  90 degree shaped pulse
705     ;p12: f1 channel - 180 degree shaped pulse
706     ;p16: homospoil/gradient pulse                   [1 msec]
707     ;p19: gradient pulse 2                          [500 usec]
708     ;p21: f3 channel -  90 degree high power pulse
709     ;p22: f3 channel - 180 degree high power pulse
710     ;p25: f1 channel - 180 degree shaped pulse (Ca, sp28)
711     ;p59: f3 channel - 180 degree shaped pulse (N, selective for Pro)
712     ;d0 : incremented delay (F1 in 3D)               [3 usec]
713     ;d1 : relaxation delay; 1-5 * T1
714     ;d3 : 1/(6J(HCa))                               [950 usec]
715     ;d4 : 1/(4J(HCa))                               [1.8 msec]
716     ;d9 : TOCSY mixing time                         [12 msec]
717     ;d10: incremented delay (F2 in 3D)              [3 usec]
718     ;d11: delay for disk I/O                        [30 msec]
719     ;d12: delay for power switching                 [20 usec]
720     ;d16: delay for homospoil/gradient recovery     [150 usec]
721     ;d22: 1/(4J(COCa))                              [4.5 msec]
722     ;d23: 1/(4J(NCO))                               [16.6 msec]
723     ;cnst21: CO chemical shift (offset, in ppm)
724     ;cnst22: Calpha chemical shift (offset, in ppm)
```

```
;cnst23: Caliphatic chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)
;l0: flag to switch between inphase and antiphase
;l1: loop for FLOPSY16 cycle: ((p6*188.448) * l1) = mixing time
;inf1: 1/SW(Cali) = 2 * DW(Cali)
;inf2: 1/SW(N) = 2 * DW(N)
;in0: 1/(2 * SW(Cali)) =  DW(Cali)
;nd0: 2
;in10: 1/(2 * SW(N)) =  DW(N)
;nd10: 2
;ns: 16 * n
;ds: >= 32
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or TPPI) in F1
;FnMODE: States-TPPI (or TPPI) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 60%
;gpz2: 50%
;gpz3: 19%
;gpz4: 11%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100


                                ;preprocessor-flags-start
;LABEL_F2: to do ipap in F2 start experiment with
;       option -DLABEL_F2 (eda: ZGOPTNS)
                                ;preprocessor-flags-end


;use AU-program splitcomb [ipap 2] (F1 or F2 with LABEL_F2) to process data


;$Id: $
```

Pulse sequence 3 – **3D (H)CBCANCO_pro**

```
;c_hcbcanco_ia3d_pro
;avance-version (20/07/28)
;(H)CbCaNCO
;3D sequence with
;   13C detected correlation for triple resonance using
;      multiple inept transfer steps
;
;      F2(Ha/b) -> F1(Ca/b,t1) -> F1(Ca)
;               -> F3(N,t2) -> F1(C=O,t3)
;
;on/off resonance 13C pulses using shaped pulses
;phase sensitive (t1)
;phase sensitive (t2)
;using constant time in t1
;using constant time in t2
;using IPAP scheme for virtual decoupling
;using selective N pulse for Pro
;(use parameterset )
;
;M.G. Murrali, A. Piai, W. Bermel, I.C. Felli & R. Pierattelli,
;   ChemBioChem 19, 1625-1629 (2018)
;W. Bermel, I. Bertini, V. Csizmok, I. C. Felli, R. Pierattelli &
;   P. Tompa, J. Magn. Reson. 198, 275-281 (2009)
;(W. Bermel, I. Bertini, I.C. Felli, R. Kuemmerle
;   & R. Pierattelli, J. Magn. Reson. 178, 56-64 (2006) )
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=


prosol relations=<triple_c>


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>


"p4=p3*2"
"p22=p21*2"
"d11=30m"
```

```
"d12=20u"

"d3=1.1m"
"d4=1.8m"
"d22=4.5m"
"d23=11.0m"
"d25=4.0m"
"d27=16.0m"


"d0=3u"
"d10=d23"
"d20=d25-p12-4u"
"d30=d27-larger(p12,p59)/2"

"in0=inf1/2"
"in10=inf2/2"

"in20=in0"
"in30=in10"

"td1=tdmax(td1,d20*2,in20)"
"td2=tdmax(td2,d30*2,in30)"


"DELTA1=d25-d0-larger(p12,p22)-d3-p4"
"DELTA2=d27-d10-p12-larger(p12,p59)/2"
"DELTA3=d27/2-p12/2"
"DELTA4=d27-d22-p12"


"l0=1"


"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst21-cnst23)/1000000)"
"spoff27=bf1*((cnst22-cnst21)/1000000)"


"o1_F1=bf1*cnst23/1000000"


aqseq 321


1 ze
```

```
 d11 pl12:f2 pl16:f3
2 d11 do:f2 do:f3
3 d12 fq=cnst23(bf ppm):f1
 d1
 d12 pl2:f2 pl3:f3
 50u UNBLKGRAD

 (p3 ph1):f2
 d4
 (center (p12:sp24 ph1) (p4 ph1):f2 )
 d4
 (p3 ph2):f2

 p16:gp1
 d16

 (p11:sp23 ph3)
 d0
 (center (p12:sp26 ph6) (p22 ph1):f3 )
 d3
 (p4 ph1):f2
 DELTA1
 (p12:sp24 ph1)
 d20
 (p12:sp26 ph1)
 4u
 (p11:sp25 ph1)

 d23
 (center (p12:sp24 ph1) (p22 ph1):f3 )
 d23 pl12:f2
 (p11:sp23 ph1)

 p16:gp2
 d16 fq=cnst21(bf ppm):f1
 20u cpd2:f2

 (p21 ph4):f3
 d10
 (p12:sp27 ph6)
 DELTA2
 (center (p12:sp24 ph6) (p59:sp46 ph1):f3 )
 d30 pl3:f3
 (p21 ph1):f3

 4u do:f2
 p16:gp3
 d16
```

```
914     20u cpd2:f2
915
916     if "l0 %2 == 1"
917       {
918       (p11:sp23 ph5)
919       DELTA3
920       (p12:sp27 ph1)
921       DELTA3
922       (center (p12:sp24 ph1) (p22 ph1):f3 )
923       DELTA3
924       (p12:sp27 ph1)
925       DELTA3 pl16:f3
926       }
927     else
928       {
929       (p11:sp23 ph7)
930       d22
931       (p12:sp27 ph1)
932       DELTA4
933       (center (p12:sp24 ph1) (p22 ph1):f3 )
934       DELTA3
935       DELTA3 pl16:f3
936       (p12:sp27 ph1)
937       }
938
939     4u BLKGRAD
940
941     go=2 ph31 cpd3:f3
942     d11 do:f2 do:f3 mc #0 to 2
943
944   #   ifdef LABEL_F1
945       F1I(iu0, 2)
946   #   else
947       F2I(iu0, 2)
948   #   endif /*LABEL_F1*/
949
950       F1PH(calph(ph3, +90), caldel(d0, +in0) & caldel(d20, -in20))
951       F2PH(calph(ph4, +90), caldel(d10, +in10) & caldel(d30, -in30))
952   exit
953
954
955   ph1=0
956   ph2=1
957   ph3=0 2
958   ph4=0 0 2 2
959   ph5=0 0 0 0 2 2 2 2
960   ph6=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
961   ph7=3 3 3 3 1 1 1 1
```

```
ph31=0 2 2 0 2 0 0 2


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl16: f3 channel - power level for CPD/BB decoupling
;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
;               for time reversed pulse
;sp26: f1 channel - shaped pulse 180 degree  (C=O off resonance)
;sp27: f1 channel - shaped pulse 180 degree  (Ca off resonance)
;sp46: f3 channel - shaped pulse 180 degree  (N, selective for Pro)
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p11: f1 channel -  90 degree shaped pulse
;p12: f1 channel - 180 degree shaped pulse
;p16: homospoil/gradient pulse                  [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p59: f3 channel - 180 degree shaped pulse (N, selective for Pro)
;d0 : incremented delay (F1 in 3D)              [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d3 : 1/(6J(HCa))                       [1.1 msec]
;d4 : 1/(4J(HCa))                       [1.8 msec]
;d10: incremented delay (F2 in 3D) = d23
;d11: delay for disk I/O                    [30 msec]
;d12: delay for power switching             [20 usec]
;d16: delay for homospoil/gradient recovery
;d20: decremented delay (F1 in 3D) = d25-p12-4u
;d22: 1/(4J(COCa))                      [4.5 msec]
;d23: 1/(4J(NCa))                       [11.0 msec]
;d25: 1/(8J'(CaCb))                     [4.0 msec]
;d27: 1/(4J(NCO))                       [16.0 msec]
;d30: decremented delay (F2 in 3D) = d27-larger(p12,p59)/2
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;cnst23: Caliphatic chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)
;l0: flag to switch between inphase and antiphase
;inf1: 1/SW(Ca) = 2 * DW(Ca)
;inf2: 1/SW(N) = 2 * DW(N)
;in0: 1/(2 * SW(Ca)) =  DW(Ca)
;nd0: 2
;in10: 1/(2 * SW(N)) =  DW(N)
;nd10: 2
```

```
;in20: = in0
;in30: = in10
;ns: 16 * n
;ds: >= 32
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or TPPI) in F1
;FnMODE: States-TPPI (or TPPI) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence


;for z-only gradients:
;gpz1: 50%
;gpz2: 30%
;gpz3: 19%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100



                                ;preprocessor-flags-start
;LABEL_F2: to do ipap in F2 start experiment with
;           option -DLABEL_F2 (eda: ZGOPTNS)
                                ;preprocessor-flags-end


;use AU-program splitcomb [ipap 2] (F1 or F2 with LABEL_F2) to process data



;$Id: $

```

Pulse sequence 4 – **3D (H)CACOCON_pro**

```
;c_hcacocon_ia3d_pro
;avance-version (20/07/28)
;(HCa)COCON-TOCSY
;3D sequence with
;   homonuclear Hartman-Hahn transfer using MOCCA-XY16
;      sequence for mixing
;
;      F2(H) -> F1(Ca) -> F1(C=O,t1, MOCCA) -> F3(N,t2) -> F1(C=O,t3)
;
;on/off resonance 13C pulses using hard and shaped pulses
;phase sensitive (t1)
;phase sensitive (t2)
;using semi-constant time in t1
;using IPAP scheme for virtual decoupling
;using selective N pulse for Pro
;(use parameterset )
;
;M.G. Murrali, A. Piai, W. Bermel, I.C. Felli & R. Pierattelli,
;   ChemBioChem 19, 1625-1629 (2018)
;I.C. Felli, R. Pierattelli, S.J. Glaser & B. Luy,
;   J. Biomol. NMR 43, 187-196 (2009)
;W. Bermel, I. Bertini, I.C. Felli, Y.-M. Lee, C. Luchinat & R.
Pierattelli,
;   J. Am. Chem. Soc. 128, 3918-3919
;S. Balayssac, B. Jimenez & M. Piccioli,
;   J. Magn. Reson. 182, 325-329 (2006)
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

prosol relations=<triple_c>

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>


"p22=p21*2"
"d11=30m"
"d12=20u"

"d3=1.1m"
"d4=1.8m"
"d22=4.5m"
```

```
"d23=16.6m"

"d21=300u"


"d10=3u"
"d28=3u"
"d0=d22"
"d20=d22+d28-4u"

"in0=inf1/2"
"in10=inf2/2"

"FACTOR1=d20*10000000*2/td1"
"INCR1=FACTOR1/10000000"

"if ( INCR1 > in0 ) { in20 = in0; } else { in20 = INCR1; }"
"if ( INCR1 > in0 ) { in28 = 0; } else { in28=in0-INCR1; }"

"TAU=d21/2"

"FACTOR2=(d9/((d21+p14))*16)"
"l1=FACTOR2*16"

"d31=(d21+p14)*l1"

"DELTA=d10*2+p8"
"DELTA1=d4-larger(p4,p25)/2"
"DELTA2=d22-d3-p4"
"DELTA3=d23/2-p12/2"
"DELTA4=d23-d22-p12"
"DELTA5=d23-larger(p12,p59)/2"

"l0=0"

"spoff13=bf1*((cnst21/2+cnst22/2)/1000000)-o1"
"spoff23=0"
"spoff24=0"
"spoff25=0"
"spoff26=bf1*((cnst22-cnst21)/1000000)"
"spoff27=bf1*((cnst21-cnst22)/1000000)"


"o1_F1=bf1*cnst21/1000000"


aqseq 321


1 ze
  d11 pl12:f2 pl16:f3
2 d11 do:f2 do:f3
```

```
    4u BLKGRAD
    d1 pl1:f1 pl2:f2
    20u rpp11
    50u UNBLKGRAD
    d12 cpd3:f3
    20u fq=cnst22(bf ppm):f1

    (p3 ph1):f2
    DELTA1
    (center (p25:sp28 ph1) (p4 ph1):f2 )
    DELTA1
    (p3 ph2):f2

    (p11:sp23 ph1)
    d3
    (p4 ph1):f2
    DELTA2
    (p12:sp27 ph1)
    4u
    (p25:sp28 ph1)
    d22
    (p12:sp27 ph1)
    4u
    (p11:sp25 ph5)

    p16:gp4
    d16 fq=cnst21(bf ppm):f1

    (p11:sp23 ph3)
    d0
    (p12:sp26 ph1)
    d28
    (p12:sp24 ph1)
    d20
    (p12:sp26 ph1)
    4u
    (p11:sp25 ph4)

    4u do:f3
    p16:gp1
    d16 pl20:f1

                                    ;begin MOCCA-XY16
  4 TAU
    (p14 ph11)
    TAU ipp11
    lo to 4 times l1
                                    ;end MOCCA-XY16

    p16:gp2
    d16 pl12:f2 pl3:f3
    20u cpd2:f2
```

```
1202
1203     (p11:sp23 ph1)
1204     DELTA5
1205     (center (p12:sp24 ph1) (p59:sp46 ph1):f3 )
1206     DELTA5
1207     (p11:sp25 ph2)
1208
1209     4u do:f2
1210     p16:gp3
1211     d16 pl3:f3
1212     20u cpd2:f2
1213
1214     (p21 ph6):f3
1215     d10
1216     (p8:sp13 ph1)
1217     d10
1218     (p22 ph1):f3
1219     DELTA
1220     (p21 ph1):f3
1221
1222     if "l0 %2 == 0"
1223        {
1224        (p11:sp23 ph1)
1225        DELTA3
1226        (p12:sp26 ph1)
1227        DELTA3
1228        (center (p12:sp24 ph1) (p22 ph1):f3 )
1229        DELTA3
1230        (p12:sp26 ph1)
1231        DELTA3 pl16:f3
1232        }
1233     else
1234        {
1235        (p11:sp23 ph7)
1236        d22
1237        (p12:sp26 ph1)
1238        DELTA4
1239        (center (p12:sp24 ph1) (p22 ph1):f3 )
1240        DELTA3
1241        DELTA3 pl16:f3
1242        (p12:sp26 ph1)
1243        }
1244
1245     4u BLKGRAD
1246
1247     go=2 ph31 cpd3:f3
1248
1249     d11 do:f2 do:f3 mc #0 to 2
1250
1251   #   ifdef LABEL_F1
1252        F1I(iu0, 2)
1253   #   else
```

```
     F2I(iu0, 2)
#   endif /*LABEL_F1*/

     F1PH(calph(ph3, +90), caldel(d0, +in0) & caldel(d20, -in20) &
caldel(d28, +in28))
     F2PH(calph(ph6, +90), caldel(d10, +in10))

  d31
exit


ph1=0
ph2=1
ph3=0 2
ph4=1 1 1 1 1 1 1 1 3 3 3 3 3 3 3 3
ph5=0 0 0 0 2 2 2 2
ph6=0 0 2 2
ph7=3
ph11=0 1 0 1 1 0 1 0 2 3 2 3 3 2 3 2
ph31=0 2 2 0 2 0 0 2 2 0 0 2 0 2 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl16: f3 channel - power level for CPD/BB decoupling
;pl20: f1 channel - power level for TOCSY (high sel.)
;sp13: f1 channel - shaped pulse 180 degree  (adiabatic)
;sp23: f1 channel - shaped pulse  90 degree  (on resonance)
;sp24: f1 channel - shaped pulse 180 degree  (on resonance)
;sp25: f1 channel - shaped pulse  90 degree  (on resonance)
;                   for time reversed pulse
;sp26: f1 channel - shaped pulse 180 degree  (Ca off resonance)
;sp27: f1 channel - shaped pulse 180 degree  (C=O off resonance)
;sp28: f1 channel - shaped pulse 180 degree  (Ca on resonance)
;                   sp28 requires higher selectivity than sp24
;sp46: f3 channel - shaped pulse 180 degree  (N, selective for Pro)
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p11: f1 channel -  90 degree shaped pulse
;p12: f1 channel - 180 degree shaped pulse
;p14: f1 channel - 180 degree low power pulse at pl20
;p16: homospoil/gradient pulse                          [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p25: f1 channel - 180 degree shaped pulse (Ca, sp28)
;p59: f3 channel - 180 degree shaped pulse (N, selective for Pro)
;d0 : incremented delay (F1 in 3D): = d22
;d1 : relaxation delay; 1-5 * T1
;d3 : 1/(6J(HCa))                                       [1.1 msec]
```

```
;d4 : 1/(4J(HCa))                                        [1.8 msec]
;d9 : TOCSY mixing time
;d10: incremented delay (F2 in 3D)                       [3 usec]
;d11: delay for disk I/O                                 [30 msec]
;d12: delay for power switching                          [20 usec]
;d16: delay for homospoil/gradient recovery
;d20: decremented delay (F1 in 3D): = d22+d28-4u
;d21: delay for MOCCA sequence                           [300 usec]
;d22: 1/(4J(COCa))                                       [4.5 msec]
;d23: 1/(4J(NCO))                                        [15.0 msec]
;d28: incremented delay (F1 in 3D)                       [3 usec]
;d31: total mixing time as executed
;cnst21: CO chemical shift (offset, in ppm)
;cnst22: Calpha chemical shift (offset, in ppm)
;o1p: CO chemical shift (cnst21)
;l0: flag to switch between inphase and antiphase
;l1: loop for MOCCA-XY16 cycle: ((TAU*2+p14)*16 * l1) = mixing time
;inf1: 1/SW(CO) = 2 * DW(CO)
;inf2: 1/SW(N) = 2 * DW(N)
;in0: 1/(2 * SW(CO)) =  DW(CO)
;nd0: 2
;in10: 1/(2 * SW(N)) =  DW(N)
;nd10: 2
;in20: = k * in0
;in28: = (1 - k) * in0
;ns: 8 * n
;ds: >= 32
;td1: number of experiments in F1
;td2: number of experiments in F2
;FnMODE: States-TPPI (or TPPI) in F1
;FnMODE: States-TPPI (or TPPI) in F2
;cpd2: decoupling according to sequence defined by cpdprg2
;cpd3: decoupling according to sequence defined by cpdprg3
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;pcpd3: f3 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1: 80%
;gpz2: 70%
;gpz3: 30%
;gpz4: 50%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100

;use AU-program splitcomb [ipap 2] to process data

;$Id: $
```